Scalaris

Users and Developers Guide
Version 0.1

Florian Schintke, Thorsten Schütt

July 24, 2008

Contents

I. Users Guide 5

1. Download and Installation 7
- 1.1. Requirements . . . 7
- 1.2. Download . . . 7
  - 1.2.1. Development Branch . . . 7
  - 1.2.2. Releases . . . 7
- 1.3. Configuration . . . 7
- 1.4. Build . . . 8
  - 1.4.1. Linux . . . 8
  - 1.4.2. Java-API . . . 8
- 1.5. Running Scalaris . . . 8
  - 1.5.1. Running on a local machine . . . 8
  - 1.5.2. Running distributed . . . 9
  - 1.5.3. Running on PlanetLab . . . 9
  - 1.5.4. Replication Degree . . . 9
  - 1.5.5. Routing Scheme . . . 9
- 1.6. Installation . . . 9

2. Using the system 11
- 2.1. Erlang . . . 11
- 2.2. Java command line interface . . . 11
- 2.3. Java API . . . 11

3. Testing the system 13

II. Developers Guide 15

4. How a node joins the system 17
- 4.1. General Erlang server loop . . . 17
- 4.2. Starting additional local nodes after boot . . . 17
  - 4.2.1. Supervisor-tree of a Scalaris node . . . 18
  - 4.2.2. Starting the or-supervisor and main processes of a node . . . 18
  - 4.2.3. Starting the and-supervisor with a peer and its local database . . . 19
  - 4.2.4. Initializing a `cs_node`-process . . . 20
  - 4.2.5. Actually joining the ring . . . 21
  - 4.2.6. Beginning to server requests . . . 23
  - 4.2.7. FAQ . . . 24

5. Directory Structure of the Source Code 25

6. System Components 27

7. Processes **29**

8. Troubleshooting **31**
8.1. ApplicationMonitor appmon:start() 31

A. Java API **33**
A.1. de.zib.chordsharp.ChordSharp 33

Part I.

Users Guide

1. Download and Installation

1.1. Requirements

For building and running Scalaris, some third-party modules are required which are not included in the Scalaris sources:

- Erlang R12
- Erlang OTP (included in Erlang R12)
- Erlang yaws
- GNU Make
- rrdtool

Note, the Version 12 of Erlang is required. Scalaris will not work with older versions.
To build the Java API the following modules are required additionally:

- Java Development Kit 1.6
- Apache Ant

Before building the Java API, make sure that **JAVA_HOME** and **ANT_HOME** are set. **JAVA_HOME** has to point to a JDK 1.6 installation, and **ANT_HOME** has to point to an Ant installation.

1.2. Download

The sources can be obtained from http://code.google.com/p/scalaris.

1.2.1. Development Branch

You find the latest development version in the svn repoitory:

```
# Non-members may check out a read-only working copy anonymously over HTTP.
svn checkout http://scalaris.googlecode.com/svn/trunk/ scalaris-read-only
```

1.2.2. Releases

Releases can be found under the 'Download' tab on the web-page.

1.3. Configuration

Scalaris is configured by two configuration files (**bin/scalaris.cfg** and **bin/scalaris.local.cfg**). It will read the former for default values and then the latter which can override the defaults. After going through the build process there will be no **scalaris.local.cfg**. It can be created by the user to alter settings. It is required to be able to run a Scalaris on distributed nodes.

File **scalaris.local.cfg**:

```
% Insert the apropriate IP-addresses for your setup as comma separated integers.
% IP Address, Port, and label of the boot server
{boot_host, {{209,85,135,99},14195,boot}}.

% IP Address, Port, and label of the log server
{log_host, {{209,85,135,99},14195,boot_logger}}.

% possible values: 14195, [14195, 14196, 14197] (list of ports),
% or {14195, 15000} as range of ports
{listen_port, 14195}.
```

boot_host defines the node where the boot server is running.

1.4. Build

1.4.1. Linux

Read the file **README** in the main Scalaris checkout directory.

1.4.2. Java-API

The following commands will build the Java API for Scalaris:

```
%> cd java-api
%> ant
```

This will build chordsharp4j.jar, which is the library for accessing the overlay network. Optionally, the documentation can be build:

```
%> ant doc
```

1.5. Running Scalaris

In Scalaris there are two kinds of processes:

- boot servers
- regular servers

In every Scalaris, at least one boot server is required. It will maintain a list of nodes taken part in the system and allows other nodes to join the ring. For redundancy, it is also possible to have several boot servers.

1.5.1. Running on a local machine

Open at least two shells. In the first, go into the bin directory:

```
%> cd bin
%> ./boot.sh
```

This will start the boot server. On success http://localhost:8000 should point to the management interface page of the boot server. The main page will show you the number of nodes currently in the system. After a couple of seconds a first Scalaris should have started in the boot server and the number should increase to one. The main page will also allow you to store and retrieve key-value pairs.

The boot server should show output similar to the following, when starting the first Scalaris nodes. The first line is printed when the Scalaris is spawned. Afterwards it will try to connect the boot server. When the third line is printed, it managed to contact the boot server and joined the ring. In this case, it was the first node in the ring.

```
[ I | Node    | <0.97.0> ] joining "23947834870"
[ I | Node    | <0.97.0> ] join as first [50,51,57,52,55,56,51,52,56,55,48]
[ I | Node    | <0.97.0> ] joined
```

In a second shell, you can now start a second Scalaris node. This will be a 'regular server'. Go in the bin directory:

```
%> cd bin
%> ./cs_local.sh
```

The second node will read the configuration file and use this information to contact the boot server and will join the ring. The number of nodes on the web page should have increased to two by now. Optionally, a third and fourth node can be started on the same machine. In a third shell:

```
%> cd bin
%> ./cs_local2.sh
```

In a fourth shell:

```
%> cd bin
%> ./cs_local3.sh
```

This will add 3 nodes to the network. The web pages at http://localhost:8000 should show the additional nodes.

1.5.2. Running distributed

Scalaris can be installed on other machines in the same way as described in Sect. **??**. Please make sure, that the **scalaris.local.cfg** refers to available boot servers. Otherwise the other nodes will not find the boot server. On the remote nodes, you only need to call **./cs_local.sh** and they will automatically contact the configured boot server.

1.5.3. Running on PlanetLab

1.5.4. Replication Degree

1.5.5. Routing Scheme

1.6. Installation

Note: there is no **make install** at the moment! The nodes have to be started from the bin directory.

2. Using the system

2.1. Erlang

2.2. Java command line interface

The jar file contains a small command line interface client.

```
%> java -jar chordsharp4j.jar -help
usage: chordsharp
 -getsubscribers <topic>   get subscribers of a topic
 -help                     print this message
 -publish <params>         publish a new message for a topic: <topic>
                           <message>
 -read <key>               read an item
 -subscribe <params>       subscribe to a topic: <topic> <url>
 -write <params>           write an item: <key> <value>
```

Read and write can be used to read resp. write from/to the overlay. getsubscribers, publish, and subscribe are the PubSub functions.

```
%> java -jar chordsharp4j.jar -write foo bar
write(foo, bar)
%> java -jar chordsharp4j.jar -read foo
read(foo) == bar
```

The chordsharp4j library requires that you are running a 'regular server' on the same node. Having a boot server running on the same node is not sufficient.

2.3. Java API

3. Testing the system

Part II.

Developers Guide

4. How a node joins the system

2008-07-22, SVN revision 1

4.1. General Erlang server loop

Servers in Erlang often use the following structure to maintain a state while processing received messages:

```
receive
  Message ->
    State1 = f(State),
    loop(State1)
end.
```

The server runs an endless loop, that waits for a message, processes it and calls itself using tail-recursion in each branch. The loop works on a **State**, which can be modified when a message is handeled.

4.2. Starting additional local nodes after boot

After booting a new Scalaris-System as described in Section 1.5.1 on page 8, ten additional local nodes can be started by typing **admin:add_nodes**(10) in the Erlang-Shell that the boot process opened.

```
scalaris/bin> ./boot.sh
Erlang (BEAM) emulator version 5.6.1 [source] [smp:2] [async-threads:4] [hipe] [kernel-poll:false]

Eshell V5.6.1  (abort with ^G)
(boot@csr-pc9)1> ["scalaris.cfg","scalaris.local.cfg"]
starting config
listening on {0,0,0,0}:{14195,15000}
[ I | Fail    | <0.75.0> ] start detector

=INFO REPORT==== 18-Jul-2008::20:08:48 ===
Yaws: Listening to 0.0.0.0:8000 for servers
 - http://localhost:8000 under ../docroot
[ I | Fail    | <0.102.0> ] start detector
this() == {{127,0,0,1},14195}
[ I | Node    | <0.105.0> ] joining "22500913918"
[ I | Node    | <0.105.0> ] join as first [50,50,53,48,48,57,49,51,57,49,56]
[ I | Node    | <0.105.0> ] joined

(boot@csr-pc9)1> admin:add_nodes(10)
```

In the following we will trace, what this function does to join additional nodes to the system. The function **admin:add_nodes**(**int**) is defined as follows.

File **admin.erl**:

```
[...]
add_nodes(Count) ->
    randoms:init(),
    add_nodes_loop(Count).
```

```
add_nodes_loop(0) ->
    ok;
add_nodes_loop(Count) ->
    io:format("~p~n", [supervisor:start_child(main_sup, {randoms:getRandomId(),
                                                         {cs_sup_or, start_link, []},
                                                         permanent,
                                                         brutal_kill,
                                                         worker,
                                                         []})]),
    add_nodes_loop(Count - 1).
[...]
```

It first initializes the random number generator and then calls **add_nodes_loop(Count)**.
This function starts a new child for the main supervisor **main_sup**. As defined by the parameters, to actually perform the start, the function **cs_sup_or:start_link** is called by the Erlang supervisor mechanism. For more details on the OTP supervisor mechanism see Chapter 18 of the Erlang book [**?**] or the online documentation at http://www.erlang.org/doc/man/supervisor.html.

4.2.1. Supervisor-tree of a Scalaris node



4.2.2. Starting the or-supervisor and main processes of a node

The supervisor mechanism first calls the **init()** function of the defined module (**cs_sup_or:init()** in this case) before calling the defined function defined start function (**start_link** here.
So, we have to take a look at **cs_sup_or:init**, the 'Scalaris *or* supervisor'.

File **cs_sup_or.erl**:

```
[...]
init([]) ->
    randoms:init(),
    InstanceId = string:concat("cs_node_", randoms:getRandomId()),
    FailureDetector =
        {failure_detector,
         {failuredetector, start_link, [InstanceId]},
         permanent, brutal_kill, worker,
         [failure_detector]
```

```
     },
    Config =
        {config,
         {config, start_link, [["scalaris.cfg", "scalaris.local.cfg"], InstanceId]},
         permanent, brutal_kill, worker,
         []},
    KeyHolder =
        {cs_keyholder,
         {cs_keyholder, start_link, [InstanceId]},
         permanent, brutal_kill, worker,
         []},
    MessageStatisticsCollector =
        {cs_message_collector,
         {cs_message, start_link, [InstanceId]},
         permanent, brutal_kill, worker,
         []},
    Supervisor_AND =
        {cs_supervisor_and,
         {cs_sup_and, start_link, [InstanceId]},
         permanent, brutal_kill, supervisor,
         []},
    {ok, {{one_for_one, 10, 1},
          [
           Config,
           KeyHolder,
           MessageStatisticsCollector,
           FailureDetector,
           Supervisor_AND
          ]}}.
[...]
```

The **init**() function of a supervised process is requested to return process descriptions that the supervisor should start, and how the processes have to be handled. Here, we define a list of processes that should be observed by a **one_for_one** supervisor. The processes are: **Config**, **KeyHolder**, **MessageStatisticsCollector** and a **Supervisor_AND** process.
The term **{one_for_one, 10, 1}** defines that the supervisor should retry 10 times to restart each process before giving up. **one_for_one** supervision means, that if a single process stops, only that process is restarted. The other processes run independently.
The **cs_sup_or:init**() is finished and the supervisor module, starts all the defined processes by calling their corresponding **init**() functions and afterwards, the functions that were defined in the list of the **cs_sup_or:init**().
For node join, we are only interested in the starting of the **Supervisor_AND** process, which was the last one in the list. All other defined processes are already running, when this will be started.

4.2.3. Starting the and-supervisor with a peer and its local database

Again, the OTP will first call the **init**() function of the corresponding module:

File **cs_sup_and.erl**:

```
[...]
init([InstanceId]) ->
    Node =
        {cs_node,
         {cs_node, start_link, [InstanceId]},
         permanent,
         brutal_kill,
         worker,
         []},
    DB =
        {cs_db_otp,
         {cs_db_otp, start_link, [InstanceId]},
         permanent,
         brutal_kill,
         worker,
```

```
          []},
    {ok, {{one_for_all, 10, 1},
          [
           DB,
           Node
          ]}}.
[...]
```

It defines two processes, that have to be observed using an **one_for_all**-supervisor, which means, that if one fails, all have to be restarted. Passed to the **init** function is the **InstanceId**, a random number to make the nodes unique. It was calculated a bit earlier in the code. Exercise: Try to find where.
As you can see from the list, the **DB** ist started before the **Node**. This is important, because **cs_node** uses the database, but not vice versa.
After the **cs_sup_and:init()** finished, the DB is initialized and afterwards **cs_node:start_link** is called by the and-supervisor.

File **cs_node.erl:**

```
[...]
start_link(InstanceId) ->
    {ok, spawn_link(?MODULE, start, [InstanceId])}.
[...]
```

This spawns a new process, that executes **cs_node:start()** and maintains a link to that process. This function is called in the and-supervisor process, so the and-supervisor process will be informed, when the spawned process finishes. After the spawn is submitted, **cs_node:start_link** returns to the supervisor, which starts observing the processes in a loop.
Note: ?**MODULE** is a predefined Erlang macro, which expands to the module name, the code belongs to (here: **cs_node**).

4.2.4. Initializing a cs_node-process

File **cs_node.erl:**

```
[...]
start(InstanceId) ->
    %register(cs_node, self()),
    process_dictionary:register_process(InstanceId, cs_node, self()),
    randoms:init(),
    timer:sleep(random:uniform(100) * 100),
    Id = cs_keyholder:get_key(),
    failuredetector:set_owner(self()),
    boot_server:connect(),
    {First, State} = cs_join:join(Id),
    if
        not First ->
            cs_replica_stabilization:recreate_replicas(
              cs_state:get_my_range(State));
        true ->
            ok
    end,
    timer:send_after(config:stabilizationInterval(),
      self(), {stabilize_ring}),
    timer:send_after(config:pointerStabilizationInterval(),
      self(), {stabilize_pointers}),
    timer:send_after(config:failureDetectorUpdateInterval(),
      self(), {stabilize_failuredetector}),
    timer:send_after(config:loadBalanceStartupInterval(),
      self(), {stabilize_loadbalance}),
    io:format("[ I | Node    | ~w ] joined~n",[self()]),
    loop(State, cs_debug:new()).
[...]
```

A **cs_node** first registers itself in the process dictionary. Then, the process sleeps for a random amount of time. Otherwise, if you would start 1000 processes with **admin:add_nodes**(1000), the boot-server would receive many join requests at the same time, which is not intended.
Then, the node retrieves its **Id** from the keyholder: **Id** = **cs_keyholder:get_key**(). In the first call, a random identifier is returned. If the **cs_node**-process failed and is restarted by its supervisor, this call to the keyholder ensures, that the node still keeps its **Id**, assuming that the keyholder process is not failing. This is important for the load-balancing and for consistent responsibility of nodes to ensure consistent lookup. Note: the name **Key-holder** actually is an id-holder.
If a node changes its position in the ring, the key-holder will be informed and the **cs_node** finishes itself. This triggers a restart of the corresponding database process via the and-supervisor. When the supervisor restarts both processes, they will retrieve the new position in the ring from the key-holder and join the ring there.
The supervisor was configured to restart a node at most 10 times. Does that mean, that a node can only change its position in the ring 10 times (caused by load-balancing)?
Next, the **cs_node** registers itself as the owner of the failuredetector **failuredetector:set_owner**(**self**())
to become informed, when the failuredetector detects failing nodes.
With **boot_server:connect**() a connection to the boot-server is established.

4.2.5. Actually joining the ring

cs_join:join is called next in **cs_node:start**().

File **cs_join.erl**:

```
[...]
join(Id) ->
    io:format("[ I | Node    | ~w ] joining ~p ~n",[self(), Id]),
    Ringsize = boot_server:number_of_nodes(),
    if
        Ringsize == 0 ->
            State = join_first(Id),
            cs_reregister:reregister(cs_state:uniqueId(State)),
            {true, State};
        true ->
            case cs_lookup:reliable_get_node(erlang:get(instance_id),
                                             Id, 60000) of
                error ->
                    join(Id);
                {ok, Succ} ->
                    State = join_ring(Id, Succ),
                    cs_reregister:reregister(cs_state:uniqueId(State)),
                    {false, State}
            end
    end.
[...]
```

The boot-server is contacted to retrieve the known number of nodes in the ring. If the ring is empty, **join_first** is called. Otherwise, **join_ring** is called.
join_first just creates a new state for a Scalaris node conisting of an empty routing table, a successorlist containing itself, itself as its predecessor, itself, its responsibility area from **Id** to **Id** (the full ring), and a load balancing schema.

File **cs_join.erl**:

```
[...]
join_first(Id) ->
    io:format("[ I | Node    | ~w ] join as first ~w ~n",[self(), Id]),
    Me = node:make(cs_send:this(), Id),
    cs_state:new(?RT:empty(Me), [Me], Me, Me, {Id, Id}, cs_lb:new()).
```

The macro ?**RT** maps to the configured routing algorithm.
The state is defined in

File **cs_state.erl**:

```
[...]
new(RT, SuccessorList, Predecessor, Me, MyRange, LB) ->
    #state{
     routingtable = RT,
     successorlist = SuccessorList,
     predecessor = Predecessor,
     me = Me,
     my_range = MyRange,
     lb=LB,
     join_time=now(),
     deadnodes = gb_sets:new(),
     trans_log = #translog{
       tid_tm_mapping = dict:new(),
       decided = gb_trees:empty(),
       undecided = gb_trees:empty()
      }
    }.
[...]
```

If a node joins an existing ring, **reliable_get_node** for the own **Id** is called in **cs_join:join()**. This lookup delivers the successor for the joining node. The node, that is currently responsible for **Id**, but which has a larger **Id** itself. If this lookup fails for some reason, it is tried again, by recursivly calling the **join()**.
What, if the **Id** is exactly the same as that of the existing node? This could lead to lookup and responsibility inconsistency? Can this be triggered by the load-balancing? This is a bug, that should be fixed!!!
Then, **cs_join:join_ring** is called:

File **cs_join.erl**:

```
[...]
join_ring(Id, Succ) ->
    io:format("[ I | Node    | ~w ] join_ring ~w ~n",[self(), Id]),
    Me = node:make(cs_send:this(), Id),
    UniqueId = node:uniqueId(Me),
    cs_send:send(node:pidX(Succ), {join, cs_send:this(), Id, UniqueId}),
    receive
        {join_response, Pred, Data, SuccList} ->
            io:format("[ I | Node    | ~w ] got pred ~w~n",[self(), Pred]),
            IsNull = node:is_null(Pred),
            if
                IsNull ->
                    failuredetector:add_node(node:uniqueId(Succ),
                      node:id(Succ), node:pidX(Succ));
                true ->
                    failuredetector:add_nodes([{node:uniqueId(Pred),
                      node:id(Pred), node:pidX(Pred)},
                      {node:uniqueId(Succ), node:id(Succ), node:pidX(Succ)}]),
                    cs_send:send(node:pidX(Pred), {update_succ, Me})
            end,
            cs_db_otp:add_data(Data),
            cs_state:new(?RT:empty(Succ), [Succ | SuccList], Pred, Me,
              {node:id(Pred), Id}, cs_lb:new())
    end.
[...]
```

First the node is initialized. Then it sends a **join** message to the successor including a reference to itself and the chosen **Id**.
The message is received by the old node in **cs_node.erl**. There exists a **{join, X}** handler.

File **cs_node.erl:**

```
[...]
  {join, Source_PID, Id, UniqueId} = _Message ->
    ?LOG("[ ~w | I | Node    | ~w ] join~n",
              [calendar:universal_time(), self()]),
    NewState = cs_join:join_request(State, Source_PID, Id, UniqueId),
    loop(NewState, ?DEBUG(cs_debug:debug(Debug, NewState, _Message)));
[...]
```

This triggers to call **join_request** on the old node.

File **cs_join.erl:**

```
[...]
join_request(State, Source_PID, Id, UniqueId) ->
    Pred = node:new(Source_PID, Id, UniqueId),
    HisData = cs_db_otp:split_data(State, Id),
    SuccList = cs_state:succ_list(State),
    cs_send:send(Source_PID, {join_response, cs_state:pred(State), HisData, SuccList}),
    failuredetector:add_node(UniqueId, Id, Source_PID),
    cs_state:update_pred(State, Pred).
[...]
```

This resets the local predecessor to the new node and splits the data. Then it sends a **join_respone** to the new node with its former predecessor, the data, it has to host, and its successorlist. Additionally, the new node will be observed with the failuredetector.
Here, the observation of the former predecessor could be deleted from the failure detector. But periodically, all known hosts are reregistered with the failure detector, and then all other nodes are thrown away from the list of nodes to observe in the failure detector.
Finally, the new predecessor is set and the **join_request()** is finished.
Back on the new node, it waits for the **join_response** message in **cs_join:join_ring()**. The next steps after the message was received from the old node are to register the predecessor, to register all new nodes in the failure detector, to send a message to the predecessor, that its successor has changed.
Finally, all data are added to the database and the succlist is build by putting the successor to the head of the successor list of the successor: **[Succ | SuccList]**.
The next step in **cs_join:join()** is to call **cs_reregister:reregister** which reregisters the node periodically with the boot server. Then **{false, State}** is returned, which means 'this was not the first node in the ring'. Now, the **cs_join:join()** is completed.

4.2.6. Beginning to server requests

cs_join:join() was called from **cs_node:start()**, which continues

File **cs_node.erl:**

```
[...]
start(InstanceId) ->
    %register(cs_node, self()),
    process_dictionary:register_process(InstanceId, cs_node, self()),
    randoms:init(),
    timer:sleep(random:uniform(100) * 100),
    Id = cs_keyholder:get_key(),
    failuredetector:set_owner(self()),
    boot_server:connect(),
    {First, State} = cs_join:join(Id),
    if
        not First ->
            cs_replica_stabilization:recreate_replicas(
              cs_state:get_my_range(State));
```

```
        true ->
            ok
    end,
    timer:send_after(config:stabilizationInterval(),
      self(), {stabilize_ring}),
    timer:send_after(config:pointerStabilizationInterval(),
      self(), {stabilize_pointers}),
    timer:send_after(config:failureDetectorUpdateInterval(),
      self(), {stabilize_failuredetector}),
    timer:send_after(config:loadBalanceStartupInterval(),
      self(), {stabilize_loadbalance}),
    io:format("[ I | Node    | ~w ] joined~n",[self()]),
    loop(State, cs_debug:new()).
[...]
```

The **cs_replica_stabilization:recreate_replicas()** function is called, which is not yet implemented. It would recreated necessary replicas that were lost due to load-balancing and node failures.
Finally all the self-management services are started with **timer:after** and the loop for request handling is started.

4.2.7. FAQ

Question: How are the routing-table, load-balancing and paxos processes started?

Answer: They are not implemented as separate Erlang processes. All the requests are handled by the loop startet by **cs_node:start()**. They could be implemented as separate processes.

5. Directory Structure of the Source Code

The directory tree of Scalaris is structured as follows:

bin	contains shell scripts needed to work with Scalaris (e.g. start the boot services, start a node, ...)
contrib	necessary third party packages
data	rrd databases and a test dataset
doc	generated erlang documentation
docroot	root directory of the integrated webserver
java-api	a java api to Scalaris
log	log files
src	contains the Scalaris source code
tests	unit tests for Scalaris
user-dev-guide	contains the sources for this document

6. System Components

7. Processes

8. Troubleshooting

8.1. ApplicationMonitor appmon:start()

A. Java API

A.1. de.zib.chordsharp.ChordSharp

```
public class ChordSharp
```

Public ChordSharp Interface.

Version: 1.1

Author: Nico Kruber, kruber@zib.de

Method Summary	
static Vector<String>	**getSubscribers**(String topic) Gets a list of subscribers of a topic.
static void	**publish**(String topic, String content) Publishes an event under a given topic.
static String	**read**(String key) Gets the value stored with the given key.
static void	**subscribe**(String topic, String url) Subscribes a url for a topic.
static void	**write**(String key, String value) Stores the given key/value pair.

read

```
public static String read(String key)
                  throws ConnectionException,
                         TimeoutException,
                         UnknownException,
                         NotFoundException
```

Gets the value stored with the given key.

Parameters: key - the key to look up

Returns: the value stored under the given key

Throws:

ConnectionException	if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie
TimeoutException	if a timeout occurred while trying to fetch the value
NotFoundException	if the requested key does not exist
UnknownException	if any other error occurs

write

```
public static void write(String key,
                         String value)
                  throws ConnectionException,
                         TimeoutException,
                         UnknownException
```

Stores the given key/value pair.

Parameters: key - the key to store the value for value - the value to store

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie TimeoutException - if a timeout occurred while trying to write the value UnknownException - if any other error occurs

publish

```
public static void publish(String topic,
                           String content)
                    throws ConnectionException
```

Publishes an event under a given topic.

Parameters: topic - the topic to publish the content under content - the content to publish

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie

subscribe

```
public static void subscribe(String topic,
                             String url)
                      throws ConnectionException
```

Subscribes a url for a topic.

Parameters: topic - the topic to subscribe the url for url - the url of the subscriber (this is where the events are send to)

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie

getSubscribers

```
public static Vector<String> getSubscribers(String topic)
                                     throws ConnectionException,
                                            UnknownException
```

Gets a list of subscribers of a topic.

Parameters: topic - the topic to get the subscribers for

Returns: the subscriber URLs

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie
UnknownException - is thrown if the return type of the erlang method does not match the expected one

de.zib.chordsharp.Transaction

```
public class \textbf{Transaction}
```

Provides means to realise a transaction with the chordsharp ring using Java.
It reads the connection parameters from a file called ChordSharpConnection.properties or uses default properties defined in ChordSharpConnection.defaultProperties.

```
OtpErlangString otpKey;
OtpErlangString otpValue;
OtpErlangString otpResult;
String key;
String value;
String result;

Transaction transaction = new Transaction(); // Transaction()
transaction.start(); // start()
transaction.write(otpKey, otpValue); // write(OtpErlangString, OtpErlangString)
transaction.write(key, value); // write(String, String)
otpResult = transaction.read(otpKey); //read(OtpErlangString)
result = transaction.read(key); //read(String)
transaction.commit(); // commit()
```

For more examples, have a look at TransactionReadExample, TransactionParallelReadsExample, TransactionWriteExample and TransactionReadWriteExample.

Attention:

If a read or write operation fails within a transaction all subsequent operations on that key will fail as well. This behaviour may particularly be undesirable if a read operation just checks whether a value already exists or not. To overcome this situation call revertLastOp() immediately after the failed operation which restores the state as it was before that operation.
The TransactionReadWriteExample example shows such a use case.

Version: 1.0

Author: Nico Kruber, kruber@zib.de

Constructor Summary
Transaction() Creates the object's connection to the chordsharp node specified in the "ChordSharpConnection.properties" file.

Method Summary	
void	**abort**() Cancels the current transaction.
void	**commit**() Commits the current transaction.
OtpErlangString	**read**(OtpErlangString key) Gets the value stored under the given key.
String	**read**(String key) Gets the value stored under the given key.
void	**revertLastOp**() Reverts the last (read or write) operation by restoring the last state.
void	**start**() Starts a new transaction by generating a new transaction log.
void	**write**(OtpErlangString key, OtpErlangString value) Stores the given key/value pair.
void	**write**(String key, String value) Stores the given key/value pair.

Transaction

```
public Transaction()
            throws ConnectionException
```

Creates the object's connection to the chordsharp node specified in the "ChordSharpConnection.properties" file.

Throws: ConnectionException - if the connection fails

start

```
public void start()
           throws ConnectionException,
                  TransactionNotFinishedException,
                  UnknownException
```

Starts a new transaction by generating a new transaction log.

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie TransactionNotFinishedException - if an old transaction is not finished (via commit() or abort()) yet UnknownException - if the returned value from erlang does not have the expected type/structure

commit

```
public void commit()
            throws UnknownException,
                   ConnectionException
```

Commits the current transaction. The transaction's log is reset if the commit was successful, otherwise it still retains in the transaction which must be successfully committed or aborted in order to be restarted.

Throws: UnknownException - If the commit fails or the returned value from erlang is of an unknown type/structure, this exception is thrown. Neither the transaction log nor the local operations buffer is emptied, so that the commit can be tried again. ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie

See Also: abort()

abort

```
public void abort()
```

Cancels the current transaction.
For a transaction to be cancelled, only the transLog needs to be reset. Nothing else needs to be done since the data was not modified until the transaction was committed.

read

```
public OtpErlangString read(OtpErlangString key)
                     throws ConnectionException,
                            TimeoutException,
                            UnknownException,
                            NotFoundException
```

Gets the value stored under the given key.

Parameters: key - the key to look up

Returns: the value stored under the given key

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie TimeoutException - if a timeout occurred while trying to fetch the value NotFoundException - if the requested key does not exist UnknownException - if any other error occurs

read

```
public String read(String key)
            throws ConnectionException,
                   TimeoutException,
                   UnknownException,
                   NotFoundException
```

Gets the value stored under the given key.

Parameters: key - the key to look up

Returns: the value stored under the given key

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie TimeoutException - if a timeout occurred while trying to fetch the value NotFoundException - if the requested key does not exist UnknownException - if any other error occurs

write

```
public void write(OtpErlangString key,
                  OtpErlangString value)
           throws ConnectionException,
                  TimeoutException,
                  UnknownException
```

Stores the given key/value pair.

Parameters: key - the key to store the value for value - the value to store

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie TimeoutException - if a timeout occurred while trying to write the value UnknownException - if any other error occurs

write

```
public void write(String key,
                  String value)
           throws ConnectionException,
                  TimeoutException,
                  UnknownException
```

Stores the given key/value pair.

Parameters: key - the key to store the value for value - the value to store

Throws: ConnectionException - if the connection is not active or a communication error occurs or an exit signal was received or the remote node sends a message containing an invalid cookie TimeoutException - if a timeout occurred while trying to write the value UnknownException - if any other error occurs

revertLastOp

```
public void revertLastOp()
```

Reverts the last (read, parallelRead or write) operation by restoring the last state. If no operation was initiated yet, this method does nothing.
This method is especially useful if after an unsuccessful read a value with the same key should be written which is not possible if the failed read is still in the transaction's log.